

今天这篇文章，我们来介绍一个带运行计时的电机控制函数块，为后续的电机调度函数块做准备。我之前曾写过一个电机控制函数块FB5001_MotorControl（[西门子SCL编程实例——带反馈监视的电机控制](#)），我们在它的基础上，增加运行时间累计功能。



打开博途新建项目，将FB5001_MotorControl添加到项目中，修改其名称为FB5007_MotorControlWithRuntime。在原函数块基础上：

①增加输入参数：GlobalTime和ResetRuntime。

- GlobalTime：来自循环中断组织块的全局时间（单位：秒）；
- ResetRuntime：复位函数块内部的用于计时的变量；

②增加输入/输出参数：

- Runtime：电机运行的实际时间（单位：秒）。多次启动的运行时间可累积；

③增加用于计时的静态变量：

- statStartTime：按下启动按钮的时间（单位：秒）；
- statLastRunningTime：上次运行的时间（单位：秒），用于累计；
- statRunTime：当前运行时间（单位：秒）；

整个函数块（FB5007_MotorControlWithRuntime）的变量声明如下图所示：

FB5007_MotorControlWithRuntime							
	名称	数据类型	默认值	保持	设定值	监控	注释
1	Input						
2	Fuse	Bool	false	非保持	<input type="checkbox"/>		fuse signal of the motor(1=ok)
3	Start	Bool	false	非保持	<input type="checkbox"/>		start button of the motor(positive edge)
4	Stop	Bool	false	非保持	<input type="checkbox"/>		stop button of the motor(positive edge)
5	FeedbackSignal	Bool	false	非保持	<input type="checkbox"/>		feedback signal of the motor(inverse sig..
6	FeedbackTime	Time	T#2s	非保持	<input type="checkbox"/>		
7	GlobalTime	UDInt	0	非保持	<input type="checkbox"/>		global time (second) from interrupt OB
8	ResetRuntime	Bool	false	非保持	<input type="checkbox"/>		reset runtime
9	Output						
10	Q	Bool	false	非保持	<input type="checkbox"/>		output signal of the motor
11	Error	Bool	false	非保持	<input type="checkbox"/>		1=error occurs;0=no error
12	Diagnose	Byte	16#0	非保持	<input type="checkbox"/>		when error=1,Diag byte give out more d...
13	InOut						
14	Runtime	UDInt	0	非保持	<input type="checkbox"/>		runtime ,can be accumulated
15	Static						
16	statStartPosEdge	Bool	false	非保持	<input type="checkbox"/>		positive edge for start input
17	statStopPosEdge	Bool	false	非保持	<input type="checkbox"/>		positive edge for stop input
18	statQ	Bool	false	非保持	<input type="checkbox"/>		static Q output
19	IEC_Timer_O_Instance	TON_TIME		非保持	<input checked="" type="checkbox"/>		
20	statStartTime	UDInt	0	非保持	<input checked="" type="checkbox"/>		start time for caculation
21	statLastRunningtime	UDInt	0	非保持	<input type="checkbox"/>		last running time
22	statRuntime	UDInt	0	非保持	<input type="checkbox"/>		unit: second
23	Temp						
24	tmpStartRisingEdge	Bool			<input type="checkbox"/>		temperable varibale for start motor
25	tmpStopRisingEdge	Bool			<input type="checkbox"/>		temperable varibale for stop motor
26	tmpCurrentTime	Time			<input type="checkbox"/>		current time value
27	tmpFeedback	Bool			<input type="checkbox"/>		
28	tmpTimerQ	Bool			<input type="checkbox"/>		
29	Constant				<input type="checkbox"/>		
30	<新增>				<input type="checkbox"/>		

代码如下:

```

1 (*
2 Copyrights @Founderchip
3 =====
4 功能说明: 带反馈监控及断路器保护监控的电机启停控制模块
5           v1.2==>增加运行时间统计功能(可累加),
6           需要使用外部的全局时间变量-GlobalTime(循环中断组织块产生)
7 输入:
8     Fuse: 电机短路其辅助触点, 1=OK;
9     Start: 电机启动信号, 1=启动;
10    Stop: 电机停止信息, 1=停止, 复位内部错误;
11    FeedbackSignal: 电机接触器或变频器的反馈信号,
12                   该信号必须在FeedbackTime时间内与Q的信号相反;
13    FeedbackTime: 反馈监控的时间, 默认2s
14    GlobalTime: 循环中断组织块产生的全局时间(秒数),
15    ResetRuntime: 复位运行时间
16 输出:
17    Q: 电机控制输出(线圈)信号;
18    Error: 功能块错误提示, 1=有错误; 0=没有错误;
19    Diagnose: 诊断字节, 当Error的值为1时, 可以从该字节获取具体的错误信息;
20 输入/输出:
21    Runtime : 实际运行时间, 可累加, reset输入信号可清零
22 诊断字节信息:
23    0=OK, 没有错误;
24    1=断路器跳闸;
25    2=反馈错误;
26    **使用停止信号复位内部错误**
27 作者:
28     北岛李工
29     2020-9-30
30 -----
31 修改日志:
32 2022-12-6 v1.2 增加运行时间统计

```

```

32 | 2022-12-0 v1.2 增加运行时间统计
33 | 2021-6-22 v1.1 增加反馈监控时间
34 | 2020-9-30 v1.0 版本(首发)          北岛李工
35 | =====
36 | *)
37 |
38 | //获取启动信号的上升沿
39 | #tmpStartRisingEdge := #Start AND NOT #statStartPosEdge;
40 | #statStartPosEdge := #Start;
41 | //获取停止信号的上升沿
42 | #tmpStopRisingEdge := #Stop AND NOT #statStopPosEdge;
43 | #statStopPosEdge := #Stop;
44 | //启动信号
45 | IF #tmpStartRisingEdge THEN
46 |     #statQ := TRUE;
47 |     //获取启动时全局时间(秒)
48 |     #statStartTime := #GlobalTime;
49 |     //获取上次运行的时间(秒)
50 |     #statLastRunningtime := #Runtime;
51 | END_IF;
52 |
53 | //停止信号
54 | IF #tmpStopRisingEdge THEN
55 |     #statQ := FALSE;
56 |     #Error := 0;
57 |     #Diagnose := 0; //清除错误代号
58 | END_IF;
59 | //断路器跳闸
60 | IF NOT #Fuse THEN
61 |     #statQ := FALSE;
62 |     #Error := 1;
63 |     #Diagnose := 1; //错误代号, 断路器跳闸
64 | END_IF;
65 |
66 | //反馈时间监控
67 | #tmpFeedback := #statQ AND #FeedbackSignal;
68 | IEC_Timer_0_Instance(IN := #tmpFeedback,
69 |                      PT := #FeedbackTime, //实际使用中该值建议修改为1~2s
70 |                      ET => #tmpCurrentTime,
71 |                      Q => #tmpTimerQ);
72 | IF #tmpTimerQ THEN
73 |     #statQ := FALSE;
74 |     #Error := 1;
75 |     #Diagnose := 2; //错误代号, 反馈监视
76 | END_IF;
77 | //运行计时(按下停止按钮后不计时, 再次启动时间可累加)
78 | IF #statQ THEN
79 |     #statRuntime := #GlobalTime - #statStartTime + #statLastRunningtime;
80 | END_IF;
81 | //运行时间清零
82 | //注意: 会把Runtime的外部地址也清零(输入/输出类型)
83 | IF #ResetRuntime THEN
84 |     #statRuntime := 0;
85 |     #statLastRunningtime := 0;
86 |     #statStartTime := #GlobalTime;
87 | END_IF;
88 | //输出
89 | #Q := #statQ;
90 | #Runtime := #statRuntime;

```

为了产生时间信号，我们添加循环中断组织块OB30，将其中断时间设置为1秒。设置一个全局变量，比如"DB200_Global".globalSecon，在OB30中将其周期性加1，代码如下：

名称	数据类型	默认值	注释
1 Input			
2 Initial_Call	Bool		Initial call of this OB
3 Event_Count	Int		Events discarded

```
1 //全局秒数
2 //每隔一秒钟加一
3 "DB200_Global".globalSecond += 1;
```

添加一个启动组织块，每次CPU重启时让全局时间变量清零（该全局时间变量的数据类型为UDInt，其取值范围为0~4_294_967_295。以每秒加1的频率计时，可支持大约136年，因此即使不清零也没有关系），如下图所示：


名称	数据类型	默认值	注释
1 Input			
2 LostRetentive	Bool		True if retentive da
3 LostRTC	Bool		True if date and tin
4 Temp			

```
1 //清零
2 "DB200_Global".globalSecond := 0;
```

这样，就可以在OB1中调用FB5007_MotorControlWithRuntime，统计其运行时间了。

程序段 1：测试

注释



我已经测试过，如果你有任何问题，欢迎留言。我们后续会在这个函数块的基础上，写电机调度的函数块（比如有多个电机，如何按照预定规则进行启停控制）。

下面是西门子SCL编程的文章归档链接：

》》 [西门子SCL编程入门到精通文章归档](#) 《《

我的书《[西门子S7-1200/1500 PLC SCL语言编程——从入门到精通](#)》从硬件到软件，比较详细的介绍了SCL语言的编程，感兴趣的话可以扫描下面的二维码查看：



识别图中小
程序码购买